

This paper describes a simple formula for calculating a smooth sinusoidal acceleration/deceleration motion profile starting with zero speed.

Given the maximum desired acceleration M and the distance D to the target position, compute the constants T , K_1 , K_2 , and K_3 as shown in the equations below:


$$T = \sqrt{\frac{2 \pi D}{M}} \quad K_1 = \frac{2 \pi}{T} \quad K_2 = \frac{M}{K_1} \quad K_3 = \frac{1}{K_1}$$

Then you can use the functions $a(t)$, $v(t)$, and $x(t)$ shown below to generate a nice smooth trajectory to the new target. "t" is elapsed time from the start of the profile:


$$\begin{aligned} a(t) &= M \sin(K_1 t) \\ v(t) &= K_2 (1 - \cos(K_1 t)) \\ x(t) &= K_2 (t - K_3 \sin(K_1 t)) \end{aligned}$$

Note that T will be the time-to-destination, $2 * K_2$ will be the maximum speed, and $M * K_1$ will be the maximum jerk.

Here's an example profile for $M=3.5 \text{ ft/s}^2$ and $D=5 \text{ ft}$:



**The plot below shows acceleration versus velocity (to check for motor saturation)
for the M and D values used for the plot on the previous page:**



If the max velocity is faster than you wanted, or if the motor(s) cannot provide the torque necessary to achieve the acceleration at the velocity values shown in the plot above, then reduce the max accel “M” and try again.

Here's a Maxima script which will display the $a(t)$, $v(t)$, and $x(t)$ functions, and plot graphs:

```
kill(all)$ assume(t>0)$

M:3.5$   D:5$   /* USER ENTERS DESIRED VALUES FOR "M" AND "D" */

/* compute constants based on M & D */
T:sqrt((2*%pi*D)/M)$
K[1]:2*%pi/T$ K[2]:M/K[1]$ K[3]:1/K[1]$

/* form and display the equations */
a(t):=' '(float(M*sin(K[1]*t)));
v(t):=' '(float(K[2]*(1-cos(K[1]*t))));
x(t):=' '(float(K[2]*(t - K[3]*sin(K[1]*t))));

/* plot a(t), v(t), and x(t) vs t */
plot2d(
[
[parametric, t, x(t), [t,0,T]],
[parametric, t, v(t), [t,0,T]],
[parametric, t, a(t), [t,0,T]]
],
[xlabel,"t"],
[style, [lines,2]],
[ylabel,"distance, speed, & accel"],
[legend, "x(t)", "v(t)", "a(t)"],
[gnuplot_preamble,
"set key top center; set xtics 0.2; set ytics 1; set grid"]
)$

/* plot accel vs velocity (to check for motor saturation) */
a: makelist (a(t),t,0,T/2,T/256)$
v: makelist (v(t),t,0,T/2,T/256)$

plot2d(
[discrete, v, a],
[xlabel,"velocity"],
[ylabel,"acceleration"],
[style, [lines,2]],
[legend, ""],
[gnuplot_preamble,"set key top center; set xtics 0.2; set ytics 0.2;
set grid"]
)$
```